

# The F# Computation Expression Zoo

Tomas Petricek<sup>1</sup> and Don Syme<sup>2</sup>

<sup>1</sup> University of Cambridge, UK

<sup>2</sup> Microsoft Research Cambridge, UK  
tp322@cam.ac.uk, dsyme@microsoft.com

**Abstract.** Many computations can be structured using abstract computation types such as monoids, monad transformers or applicative functors. Functional programmers use those abstractions directly while mainstream languages often integrate concrete instances as language features – e.g. generators in Python or asynchronous computations in C# 5.0. The question is, is there a sweet spot between convenient, hardwired language features, and an inconvenient but flexible libraries?

F# *computation expressions* answer this question in the affirmative. Unlike the “do” notation in Haskell, computation expressions are not tied to a single kind of abstraction. They support a wide range of computations, depending on what operations are available. They also provide greater syntactic flexibility leading to a more intuitive syntax, without resorting to full macro-based meta-programming.

We show that computation expressions can structure well-known computations including monoidal list comprehensions, monadic parsers, applicative formlets and asynchronous sequences based on the list monad transformer. We also present typing rules for computation expressions that are capable of capturing all these applications.

## 1 Introduction

Computations with non-standard aspects like non-determinism, effects, asynchronicity or their combinations can be captured using a variety of abstract computation types. In Haskell, we write such computations using a mix of combinators and syntactic extensions like monad comprehensions [5] and “do” notation. Languages such as Python and C# emphasize the syntax and provide single-purpose support e.g. for asynchrony [1] and list generators.

Using such abstractions can be made simpler and more intuitive if we employ a general syntactic machinery. F# computation expressions provide *uniform* syntax that supports monoids, monads [23], monad transformers [11] and applicative functors [14]. They reuse familiar syntax including loops and exception handling – the laws of underlying abstractions guarantee that these constructs preserve intuition about code. At the same time, the mechanism is *adaptable* and enables appropriate syntax depending on the abstraction.

Most languages, including Haskell, Scala, C#, JavaScript and Python have multiple syntactic extensions that improve computational expressivity: queries, iterators, comprehensions, asynchronous computations are just a few. However,

“syntactic budget” for such extensions is limited. Haskell already uses three notations for comprehensions, monads and arrows [16]. C# and Scala have multiple notations for queries, comprehensions, asynchronicity and iterators. The more we get with one mechanism, the better. As we show, computation expressions give a lot for relatively low cost – notably, without resorting to full-blown macros.

Some of the technical aspects of the feature have been described before<sup>3</sup> [21], but this paper is novel in that it relates the mechanism to well-known abstract computation types. We also present new typing rules based on those uses.

**Practical examples.** We demonstrate the breadth of computations that can be structured using F# computation expressions. The applications include asynchronous workflows and sequences §2.1, §2.3, list comprehensions and monadic parsers §2.2 and formlets for web programming §2.4.

**Abstract computations.** We show that the above examples fit well-known types of abstract computations, including additive monads and monad transformers, and we show that important syntactic equalities hold as a result §4.

**Syntax and typing.** We give typing rules that capture idiomatic uses of computation expressions §3.2, extend the translation to support applicative functors §2.4 and discuss the treatment of effects §3.4 that is needed in impure languages.

We believe that software artifacts in programming language research matter [10], so all code can be run at: <http://tryjoinads.org/computations>. The syntax for applicative functors is a reserch extension; other examples require F# 2.0.

## 2 Computation expressions by example

Computation expressions are blocks of code that represent computations with a non-standard aspect such as laziness, asynchronicity, state or other. The code inside the block is re-interpreted using a *computation builder*, which is a record of operations that define the semantics, but also syntax available in the block.

Computation expressions mirror the standard F# syntax (let binding, loops, exception handling), but support additional computational constructs. For example `let!` represents the computational (monadic) alternative of let binding.

We first introduce the syntax and mapping to the underlying operations informally, but both are made precise later §3. Readers unfamiliar with F# may find additional explanation in previous publications [21]. To show the breadth of applications, we look at five examples arising from different abstractions.

### 2.1 Monadic asynchronous workflows

Asynchronous workflows [20] allow writing non-blocking I/O using a mechanism based on the *continuation monad* (with error handling etc.) The following example compares F# code with an equivalent in C# using a single-purpose feature:

<sup>3</sup> F# 3.0 extends the mechanism further to accomodate extensible query syntax. To keep this paper focused, we leave analysis of these extensions to future work.

```

let getLength url = async {
  let! html = fetchAsync url
  do! Async.Sleep 1000
  return html.Length
}
async Task<string> GetLength(string url) {
  var html = await FetchAsync(url);
  await Task.Delay(1000);
  return html.Length;
}

```

Both functions return a computation that expects a *continuation* and then downloads a given URL, waits one second and passes content length to the continuation. The C# version uses the built-in `await` keyword to represent non-blocking waiting. In F#, the computation is enclosed in the `async {...}` block, where `async` is an identifier that refers to the computation builder.

The computation builder `async` determines which of the pre-defined keywords are allowed in the computation block. The `let!` keyword represents (monadic) composition and requires the *Bind* operation. This operation also enables the `do!` keyword which is equivalent to using `let!` on a computation that returns a value of type `unit`. Finally, the `return` keyword is mapped to the *Return* operation:

```

async.Bind(fetchAsync(url), fun html →
  async.Bind(Async.Sleep 1000, fun () →
    async.Return(html.Length)))

```

The *Bind* and *Return* operations form a monad. As usual, *Return* has a type  $\alpha \rightarrow A\alpha$  and the required type of *Bind* is  $A\alpha \times (\alpha \rightarrow A\beta) \rightarrow A\beta$  (we write  $\alpha, \beta$  for universally qualified type variables and  $\tau$  as for concrete types) <sup>4</sup>.

**Sequencing and effects.** Effectful expressions in F# return a value `()` which is the only value of type `unit`. Assuming  $e_1$  has a type `unit`, we can sequence expression using  $e_1; e_2$ . We can also write effectful if condition without the `else` clause (which implicitly returns the unit value `()` in the `false` case). Both have an equivalent computation expression syntax:

```

async { if delay then do! Async.Sleep(1000)
  printfn "Starting..."
  return! asyncFetch(url) }

```

If *delay* is true, the workflow waits one second before downloading the page and returning it. The translation uses additional operations – *Zero* represents monadic unit value, *Combine* corresponds to the “;” operator and *Delay* embeds an effectful expression in a (delayed) computation. For monads, these can be defined in terms of *Bind* and *Return*, but this is not the case for all computations (e.g. monoidal computations discussed in §2.2 require different definitions).

We also use the `return!` keyword, which returns the result of a computation and requires an operation *ReturnFrom* of type  $A\alpha \rightarrow A\alpha$ . This is typically implemented as an identity function – its main purpose is to enable the `return!` keyword in the syntax, as this may not be always desirable.

<sup>4</sup> For the purpose of this paper, we write type application using a light notation  $T\tau$ .

```

async.Combine
  ( ( if delay then async.Bind(Async.Sleep(1000), fun () → async.Zero())
    else async.Zero() ), async.Delay(fun() →
      printfn "Starting..."
      async.ReturnFrom(asyncFetch(url))))

```

*Zero* has a type  $\text{unit} \rightarrow A\text{unit}$  and is inserted when a computation does not return a value, here in both branches of **if**. A computation returning `unit` can be composed with another using *Combine* which has a type  $A\text{unit} \times A\alpha \rightarrow A\alpha$  and corresponds to “;”. It runs the left-hand side before returning the result of the right-hand side. Finally, *Delay*, of type  $(\text{unit} \rightarrow A\tau) \rightarrow A\tau$ , is used to wrap any effectful computations (like printing) in the monad to avoid performing the effects before the first part of sequential computation is run.

## 2.2 Additive parsers and list comprehensions

Parsers [9] or list comprehensions differ in that they may return multiple values. Such computations can be structured using additive monads (`MonadPlus` in Haskell). These abstractions can be used with `F#` computation expressions too. Interestingly, they require different typing of *Zero* and *Combine*.

**Monadic parsers.** For parsers, we use the same notation as previously. The difference is that we can now use `return` and `return!` repeatedly. The following parsers recognize one or more and zero or more repetitions of a given predicate:

```

let rec zeroOrMore p = parse {
  return! oneOrMore p
  return [] }
and oneOrMore p = parse {
  let! x = p
  let! xs = zeroOrMore p
  return x :: xs }

```

The `oneOrMore` function uses just the monadic interface and so its translation uses *Bind* and *Return*. The `zeroOrMore` function is more interesting – it combines a parser that returns one or more occurrences with a parser that always succeeds and returns an empty list. This is achieved using the *Combine* operation:

```

let rec zeroOrMore p = parse.Delay(fun () →
  parse.Combine( parse.ReturnFrom(oneOrMore p),
    parse.Delay(fun() → parse.Return([]) )))

```

Here, *Combine* represents the monoidal operation on parsers (either left-biased or non-deterministic choice) and it has a type  $P\alpha \times P\alpha \rightarrow P\alpha$ . Accordingly, the *Zero* operation is the unit of the monoid. It has a type  $\text{unit} \rightarrow P\alpha$ , representing a parser that returns no  $\alpha$  values (rather than returning a single `unit` value).

For effectful sequencing of monads, it only makes sense to use unit-returning computations in the left-hand side of *Combine* and as the result of *Zero*. However, if we have a monoidal computation, we can define *Combine* that combines multiple produced values. This shows that the computation expression mechanism needs certain flexibility – the translation is the same, but the typing differs.

**List comprehensions.** Although list comprehensions implement the same abstract type as parsers, it is desirable to use different syntax if we want to make the syntactic sugar comparable to sbuilt-in features in other languages. The following shows an F# list comprehension and a Python generator side-by-side:

```
seq { for n in list do      for n in list :
      yield n              yield n
      yield n * 10 }      yield n * 10
```

The computations iterate over a source list and produce two results for each input. Monad comprehensions [5] allow us to write  $[ n * 10 \mid n \leftarrow list ]$  to multiply all elements by 10, but they are not expressive enough to capture the duplication. Doing that requires rewriting the code using combinators.

Although the F# syntax looks different to what we have seen so far, it is actually very similar. The `for` and `yield` constructs are translated to *For* and *Yield* operations which have the same form as *Bind* and *Return*, but provide backing for a different syntax. The translation looks as follows:

```
seq.For(list, fun () →
  seq.Combine(seq.Yield(n), seq.Delay(fun () → seq.Yield(n * 10))) )
```

*Combine* concatenates multiple results and has the standard monoidal type  $[\alpha] \times [\alpha] \rightarrow [\alpha]$ . *For* has the type of monadic bind  $[\alpha] \rightarrow (\alpha \rightarrow [\beta]) \rightarrow [\beta]$  and *Yield* has a type of monadic unit  $\alpha \rightarrow [\alpha]$ . We could have provided the *Bind* and *Return* operations in the `seq` builder instead, but this leads to a less intuitive syntax that requires users to write `let!` for iteration and `return` for yielding.

As the Python comparison shows, the flexibility of computation expressions means that they are often close to a built-in syntax. The author of a concrete computation (`parse`, `seq`, `async`, ...) chooses the appropriate syntax. For additive monads, the choice can be made based on the laws that hold §4.2.

### 2.3 Layered asynchronous sequences

It is often useful to combine non-standard aspects of multiple computations. This is captured by monad transformers [11]. Although F# does not support higher-kinded types, monad transformers still provide a useful conceptual framework.

For example, *asynchronous sequences* [17] combine non-blocking asynchronous execution with the ability to return multiple results – a file download can then produce data in 1kB buffers as they become available. Using `Async<T>` as the base type, we can follow the list monad transformer [7] and define the type as:

```
type AsyncSeqInner T = AsyncNil | AsyncCons of T * Async T
type AsyncSeq T      = Async (AsyncSeqInner T)
```

When given a continuation, an asynchronous sequence calls it with either `AsyncNil` (the end of the sequence) or with `AsyncCons` that carries a value together with the tail of the asynchronous sequence. The flexibility of computation expression makes it possible to provide an elegant syntax for writing such computations:

```

let rec urlPerSecond  $n = \text{asyncSeq}$  {
  do! Async.Sleep 1000
  yield getUrl  $i$ 
  yield! iterate ( $i + 1$ ) }
let pagePerSecond  $urls = \text{asyncSeq}$  {
  for  $url$  in urlPerSecond 0 do
    let!  $html = \text{asyncFetch } url$ 
  yield  $url, html$  }

```

The `urlPerSecond` function creates an asynchronous sequence that produces one URL per second. It uses `bind` (`do!`) of the asynchronous workflow monad to wait one second and then composition of asynchronous sequences, together with `yield` to produce the next URL. The `pagePerSecond` function uses `for` to iterate over (`bind on`) an asynchronous sequence and then `let!` to wait for (`bind on`) an asynchronous workflow. The `for` loop is asynchronous and lazy – its body is run each time the caller asks for the next result.

Asynchronous sequences form a monad and so we could use the standard notation for monads with just `let!` and `return`. We would then need explicit lifting function that turns an asynchronous workflow into an asynchronous sequence that returns a single value. However, `F#` computation expressions allow us to do better. We can define both `For` and `Bind` with the following types:

```

asyncSeq.For : AsyncSeq  $\alpha \rightarrow (\alpha \rightarrow \text{AsyncSeq } \beta) \rightarrow \text{AsyncSeq } \beta$ 
asyncSeq.Bind : Async  $\alpha \rightarrow (\alpha \rightarrow \text{AsyncSeq } \beta) \rightarrow \text{AsyncSeq } \beta$ 

```

We omit the translation of the above example – it is a straightforward variation on what we have seen so far. A more important point is that we can benefit from the fact that operations of the computation builder are not restricted to a specific type (such as `Bind` to represent binding on some monad).

As previously, the choice of the syntax is left to the author of the computation. Here, asynchronous sequences are an additive monad and so we use `for/yield`. Underlying asynchronous workflows are just monads, so it makes sense to add `let!` that automatically lifts a workflow to an asynchronous sequence.

An important aspect of the fact that asynchronous sequences can be described using a monad transformer is that certain laws hold. In §4.3 we show how these map to the computation expression syntax.

## 2.4 Applicative formlets

*Applicative functors* [14,12] are weaker (and thus more common) abstraction than monads. The difference between applicative and monadic computations is that a monadic computation can perform different effects depending on values obtained earlier during the computation. On the other hand, the effects of an applicative computation are fully determined by its structure.

In other words, it is not possible to choose which computation to run (using `let!` or `do!`) based on values obtained in previous `let!` bindings. The following example demonstrates this using a web form abstraction called formlets [2]:

```

formlet { let!  $name = \text{Formlet.textBox}$ 
         and  $gender = \text{Formlet.dropDown}$  ["Male"; "Female"]
         return  $name + " " + gender$  }

```

The computation describes two aspects – the rendering and the processing of entered values. The rendering phase uses the fixed structure to produce HTML with text-box and drop-down elements. In the processing phase, the values of *name* and *gender* are available and are used to calculate the result of the form.

The structure of the form needs to be known without having access to specific values. The syntax uses parallel binding (**let!...and...**), which binds a fixed number of independent computations. The rest of the computation cannot contain other (applicative) bindings.

There are two equivalent definitions of applicative functors. We need two operations known from the less common style. *Merge* of type  $F\alpha \times F\beta \rightarrow F(\alpha \times \beta)$  represents composition of the structure (without considering specific values) and *Map* of type  $F\alpha \times (\alpha \rightarrow \beta) \rightarrow F\beta$  transforms the (pure) value. The computation expression from the previous example is translated as follows:

```
formlet.Map
  ( formlet.Merge(Formlet.textBox, Formlet.dropDown ["Male"; "Female"]),
    fun (name, gender) → name + " " + gender )
```

The computations composed using parallel binding are combined using *Merge*. In formlets, this determines the structure used for HTML rendering. The rest of the computation is turned into a pure function passed to *Map*. Note that the translation allows uses beyond applicative functors. The **let!...and...** syntax can also be used with monads to write zip comprehensions [5].

Applicative functors were first introduced to support *applicative* programming style where monads are not needed. The *idiom brackets* notation [14] fits that purpose better. We find that computation expressions provide a useful alternative for more complex code and fit better with the impure nature of F#.

### 3 Semantics of computation expressions

The F# language specification [21] documents computation expressions as a purely syntactic mechanism. They are desugared before type-checking, which is then performed on the translated code using standard F# typing rules. Similarly to Haskell's rebindable syntax, but to a greater level, this provides flexibility that allows the users to invent previously unforeseen abstractions.

In this paper, we relate computation expressions to standard abstract computation types. In this section, we present new typing rules that capture such common uses and make the system more robust by supporting better error messages and disallowing uncommon (likely erroneous) uses.

#### 3.1 Syntax

The full syntax of computation expressions is given in the language specification, but the following lists all important constructs that we consider in this paper:

$expr = expr \{ cexpr \}$	(computation expression)
$binds = v = expr$	(single binding)
$  v = expr \text{ and } binds$	(parallel binding)

$cexpr = \mathbf{let} \ v = expr \ \mathbf{in} \ cexpr$	(binding value)
$\mathbf{let!} \ binds \ \mathbf{in} \ cexpr$	(binding computation)
$\mathbf{for} \ v \ \mathbf{in} \ expr \ \mathbf{do} \ cexpr$	(for loop computation)
$\mathbf{return} \ expr$	(return value)
$\mathbf{return!} \ expr$	(return computation)
$\mathbf{yield} \ expr$	(yield value)
$\mathbf{yield!} \ expr$	(yield computation)
$cexpr_1; cexpr_2$	(compose computations)
$expr$	(effectful expression)

We omit **do!** which can be easily expressed using **let!** To accommodate the applicative syntax, *binds* is used to express one or more parallel variable bindings.

For space reasons, we also omit imperative **while** and exception handling constructs, but both of these are an important part of computation expressions. They allow taking existing code and wrapping it in a computation block to augment it with non-standard computational aspect.

### 3.2 Typing

The Figure 1 uses three judgments. Standard F# expressions are typed using  $\Gamma \vdash expr : \tau$ . Computation expressions always return computation of type  $M\tau$  and are typed using  $\Gamma \Vdash_{\sigma} cexpr : M\tau$ . A helper judgement  $\Gamma \triangleright_{\sigma} binds : M\Sigma$  checks bindings of multiple computations and produces a variable context with newly bound variables, wrapped in the type  $M$  of the bound computations.

The latter two are parameterized by the type of the computation expression builder (such as **seq** or **async**). The operations supported by the builder determine which syntactic constructs are enabled. Typing rules that require a certain operation have a side-condition on the right, which specifies the requirement.

In most of the side-conditions, the functions are universally quantified over the type of values (written as  $\alpha, \beta$ ). This captures the fact that computation should not restrict the values that users can work with. However, this is not the case in the rules (*seq*) and (*zero*). Here, we can only require that a specific instantiation is available – the reason is that these operations may be used in two different ways. As discussed in §2.1, for monads the result of *Zero* and the first argument of *Combine* are restricted to  $M \text{unit}$ . They can be universally quantified only if the computation is monoidal §2.2.

Another notable aspect of the typing is that a single computation expression may use multiple computation types (written  $M, N, L$  and  $D$ ). In *Bind* and *For*, the type of bound argument is  $M$ , but the resulting computation is  $N$  (we require that **bind** returns the same type of computation as the one produced by the function). This corresponds to the typing used by computations arising from monad transformers §2.3. Although combining multiple computation types is not as frequent, computations often have a delayed version which we write as  $D$ . This is an important consideration for impure languages such as F# §3.4.

Finally, we omitted typing for **yield** and **yield!** because it is similar to the typing of **return** and **return!** (using *Yield* and *YieldFrom* operations, respectively).



$$\begin{array}{c}
\boxed{\Gamma \vdash expr : \tau} \quad \text{and} \quad \boxed{\Gamma \triangleright_{\sigma} binds : M\Sigma} \\
\\
(\text{run}) \frac{\Gamma \vdash expr : \sigma \quad \Gamma \Vdash_{\sigma} cexpr : M\tau}{\Gamma \vdash expr \{ cexpr \} : N\tau} \quad (\forall \alpha : \sigma. Run : D\alpha \rightarrow N\alpha \\ \forall \alpha : \sigma. Delay : (\text{unit} \rightarrow M\alpha) \rightarrow D\alpha) \\
\\
(\text{bind-one}) \frac{\Gamma \vdash expr : M\tau}{\Gamma \triangleright_{\sigma} v = expr : M(v:\tau)} \\
\\
(\text{bind-par}) \frac{\Gamma \vdash expr : \tau \quad \Gamma \triangleright_{\sigma} binds : M\Sigma}{\Gamma \triangleright_{\sigma} v = expr \text{ and } binds : M(\Sigma, v:\tau)} \quad (\forall \alpha, \beta : \sigma. Merge : \\ M\alpha \rightarrow M\beta \rightarrow M(\alpha \times \beta)) \\
\\
\boxed{\Gamma \Vdash_{\sigma} cexpr : M\tau} \\
\\
(\text{let}) \frac{\Gamma \vdash expr : \tau_1 \quad \Gamma, v:\tau_1 \Vdash_{\sigma} cexpr : M\tau_2}{\Gamma \Vdash_{\sigma} \text{let } v = expr \text{ in } cexpr : M\tau_2} \\
\\
(\text{bind}) \frac{\Gamma \triangleright_{\sigma} binds : M\Sigma \quad \Gamma, \Sigma \Vdash_{\sigma} cexpr : N\tau}{\Gamma \Vdash_{\sigma} \text{let! } binds \text{ in } cexpr : N\tau} \quad (\forall \alpha, \beta : \sigma. Bind : \\ M\alpha \rightarrow (\alpha \rightarrow N\beta) \rightarrow N\beta) \\
\\
(\text{map}) \frac{\Gamma \triangleright_{\sigma} binds : M\Sigma \quad \Gamma, \Sigma \vdash expr : \tau}{\Gamma \Vdash_{\sigma} \text{let! } binds \text{ in return } expr : N\tau} \quad (\forall \alpha, \beta : \sigma. Map : \\ M\alpha \rightarrow (\alpha \rightarrow \beta) \rightarrow N\beta) \\
\\
(\text{for}) \frac{\Gamma \vdash expr : M\tau_1 \quad \Gamma, v:\tau_1 \Vdash_{\sigma} cexpr : N\tau_2}{\Gamma \Vdash_{\sigma} \text{for } v \text{ in } expr \text{ do } cexpr : N\tau_2} \quad (\forall \alpha, \beta : \sigma. For : \\ M\alpha \rightarrow (\alpha \rightarrow N\beta) \rightarrow N\beta) \\
\\
(\text{return-val}) \frac{\Gamma \vdash expr : \tau}{\Gamma \Vdash_{\sigma} \text{return } expr : M\tau} \quad (\forall \alpha : \sigma. Return : \alpha \rightarrow M\alpha) \\
\\
(\text{return-comp}) \frac{\Gamma \vdash expr : M\tau}{\Gamma \Vdash_{\sigma} \text{return! } expr : N\tau} \quad (\forall \alpha : \sigma. ReturnFrom : M\alpha \rightarrow N\alpha) \\
\\
(\text{seq}) \frac{\Gamma \Vdash_{\sigma} cexpr_1 : M\tau_1 \quad \Gamma \Vdash_{\sigma} cexpr_2 : N\tau_2}{\Gamma \Vdash_{\sigma} cexpr_1; cexpr_2 : L\tau_1} \quad (\forall \alpha : \sigma. Delay : (\text{unit} \rightarrow N\alpha) \rightarrow D\alpha \\ \forall \alpha : \sigma. Combine : M\tau_1 \rightarrow D\alpha \rightarrow L\alpha) \\
\\
(\text{zero}) \frac{\Gamma \vdash expr : \text{unit}}{\Gamma \Vdash_{\sigma} expr : M\tau} \quad (\sigma. Zero : \text{unit} \rightarrow M\tau)
\end{array}$$

Fig. 1. Typing rules for computation expressions

### 3.3 Translation

The translation is defined as a relation  $\llbracket - \rrbracket_m$  that is parameterized by a variable  $m$  which refers to the current instance of a computation builder. This parameter is used to invoke members of the builder, such as  $m.Return(\dots)$ . Multiple variable bindings are translated using  $\langle\langle binds \rangle\rangle_m$  and we define a helper mapping  $\langle binds \rangle$  that turns bindings into a pattern that can be used to decompose a tuple constructed by merging computations using the *Merge* operation.

It is easy to check that our typing guarantees that a well-typed computation expression is always be translated to a well-typed F# expression. The side-conditions ensure that all operations are available and have an appropriate type.

$$\begin{aligned}
\text{expr } \{ \text{cexpr} \} &= \text{let } m = \text{expr} \text{ in } m.\text{Run}(m.\text{Delay}(\text{fun } () \rightarrow \llbracket \text{cexpr} \rrbracket_m)) \\
\llbracket \text{let } v = \text{expr} \text{ in } \text{cexpr} \rrbracket_m &= \text{let } v = \text{expr} \text{ in } \llbracket \text{cexpr} \rrbracket_m \\
\llbracket \text{let! binds in cexpr} \rrbracket_m &= m.\text{Bind}(\llbracket \text{binds} \rrbracket_m, \text{fun } \langle \text{binds} \rangle \rightarrow \llbracket \text{cexpr} \rrbracket_m) \\
\llbracket \text{let! binds in return expr} \rrbracket_m &= m.\text{Map}(\llbracket \text{binds} \rrbracket_m, \text{fun } \langle \text{binds} \rangle \rightarrow \text{expr}) \\
\llbracket \text{for } v \text{ in } \text{expr} \text{ do } \text{cexpr} \rrbracket_m &= m.\text{For}(\text{expr}, \text{fun } () \rightarrow \llbracket \text{cexpr} \rrbracket_m) \\
\llbracket \text{return expr} \rrbracket_m &= m.\text{Return}(\text{expr}) \\
\llbracket \text{return! expr} \rrbracket_m &= m.\text{ReturnFrom}(\text{expr}) \\
\llbracket \text{cexpr}_1; \text{cexpr}_2 \rrbracket_m &= m.\text{Combine}(\llbracket \text{cexpr}_1 \rrbracket_m, m.\text{Delay}(\text{fun } () \rightarrow \llbracket \text{cexpr}_2 \rrbracket_m)) \\
\llbracket \text{expr} \rrbracket_m &= \text{expr}; m.\text{Zero}() \\
\langle\langle v = \text{expr} \rangle\rangle_m &= \text{expr} \\
\langle\langle v = \text{expr} \text{ and binds} \rangle\rangle_m &= m.\text{Merge}(\text{expr}, \llbracket \text{binds} \rrbracket_m) \\
\langle v = \text{expr} \rangle &= v \\
\langle v = \text{expr} \text{ and binds} \rangle &= v, \langle \text{binds} \rangle
\end{aligned}$$

Fig. 2. Translation rules for computation expressions

Some readers have already noticed that our definition of  $\llbracket - \rrbracket_m$  is ambiguous. The `let!` binding followed by `return` can be translated in two different ways. We intentionally do not specify the behaviour in this paper – the laws in §4.2 require the two translations to be equivalent. For monads, this equivalence is easy to see by considering the definition of *Map* in terms of *Bind* and *Return*.

In earlier discussion, we omitted the *Run* and *Delay* members in the translation of `expr { cexpr }`. The next section discusses these two in more details.

### 3.4 Delayed computations

We already mentioned that side-effects are an important consideration when adding sequencing to monadic computations §2.1. In effectful languages, we need to distinguish between two types of monads. We use the term *monadic computation* for monads that represent a delayed computation such as asynchronous workflows or lazy list comprehensions; the term *monadic containers* will be used for monads that represent a wrapped non-delayed value (such as the option type, non-lazy list or the identity monad).

**Monadic computations.** The defining feature of *monadic computations* is that they permit a *Delay* operation of type  $(\text{unit} \rightarrow M\alpha) \rightarrow M\alpha$  that does not perform the effects associated with the function argument. For example, in asynchronous workflows, the operation builds a computation that waits for a continuation – and so the effects are only run when the continuation is provided.

Before going further, we revisit the translation of asynchronous workflows using the full set of rules to show how *Run* and *Delay* are used. Consider the the following simple computation with a corresponding translation:

<pre> <b>let</b> answer = <b>async</b> {   printfn "Welcome..."   <b>return</b> 42 } </pre>	<pre> <b>let</b> answer = <b>async</b>.Run(<b>async</b>.Delay(<b>fun</b> () →   printfn "Welcome..."   <b>async</b>.Return(42) )) </pre>
---	--

For monadic computations such as asynchronous workflows, we do not expect that defining `answer` will print “Welcome”. This is achieved by the wrapping the specified computation in the translation rule for the `expr { cexpr }` expression.

In this case, the result of `Delay` is a computation `A int` that encapsulates the delayed effect. For monadic computations, `Run` is a simple identity – contrary to what the name suggests, it does not run the computation (although that might be an interesting use beyond standard abstract computations). The need for `Run` becomes obvious when we look at monadic containers.

**Effects and monadic containers.** For monadic containers, it is impossible to define a `Delay` operation that does not perform the effects and has a type  $(\text{unit} \rightarrow M\alpha) \rightarrow M\alpha$ , because the resulting type has no way of capturing unevaluated code. However, the (*seq*) typing rule in Figure 1 permits an alternative typing. Consider the following example using the Maybe (option) monad:

```
maybe { if b = 0 then return! None
         printfn "Calculating..."
         return a / b }
```

Using the same translation rules, `Run`, `Delay` and `Delay` are inserted as follows:

```
maybe.Run(maybe.Delay(fun () → maybe.Combine
  ( (if b = 0 then maybe.ReturnFrom(None) else maybe.Zero()),
    maybe.Delay(fun () → printfn "Calculating..."
                 maybe.Return(a / b)) ) ) )
```

The key idea is that we can use two different types –  $M\alpha$  for values representing (evaluated) monadic containers and  $\text{unit} \rightarrow M\alpha$  for delayed computations. The operations then have the following types:

```
Delay    : (unit → Mα) → (unit → Mα)
Run      : (unit → Mα) → Mα
Combine  : M unit → (unit → Mα) → Mα
```

Here, the `Delay` operation becomes just an identity that returns the function created by the translation. In the translation, the result of `Delay` can be passed either to `Run` or as the second argument of `Delay`, so these need to be changed accordingly. The `Run` function now becomes important as it turns the delayed function into a value of the expected type  $M\alpha$  (by applying it).

**Unified treatment of effects.** In the typing rules §3.2, we did not explicitly list the two options, because they can be generalized. We require that the result of `Delay` is some (possibly different) abstract type  $D\alpha$  representing delayed computations. For monadic computations, the type is just  $M\alpha$  and for monadic containers, it is  $\text{unit} \rightarrow M\alpha$ . Our typing is even more flexible, as it allows usage of multiple different computation types – but treatment of effects is one example where this additional flexibility is necessary.

Finally, it should be noted that we use a slight simplification. The actual F# implementation does not strictly require `Run` and `Delay` in the translation of `expr { cexpr }`. They are only used if they are present.

## 4 Computation expression laws

Although computation expressions are not tied to any specific abstract computation type, we showed that they are usually used with well-known abstractions. This means three good things. First, we get better understanding of what computations can be encoded (and how). Second, we can add a more precise typing §3.2. Third, we know that certain syntactic transformations (refactorings) preserve the meaning of computation. This section looks at the last point.

To keep the presentation in this section focused, we assume that there are no untracked side-effects (such as I/O) and we ignore *Run* and *Delay*.

### 4.1 Monoid and semigroup laws

We start from the simplest structures. A semigroup  $(S, \circ)$  consists of a set  $S$  and a binary operation  $\circ$  such that  $a \circ (b \circ c) = (a \circ b) \circ c$ . A computation expression corresponding to a semigroup defines only *Combine* (of type  $M\alpha \times M\alpha \rightarrow M\alpha$ ). To allow appropriate syntax, we also add *YieldFrom* which is just the identity function (with a type  $M\alpha \rightarrow M\alpha$ ). The associativity implies the following syntactic equivalence:

$$m \{ cexpr_1; cexpr_2; cexpr_3 \} \equiv m \{ \mathbf{yield!} \ m \{ cexpr_1; cexpr_2 \}; cexpr_3 \}$$

A monoid  $(S, \circ, \epsilon)$  is a semigroup  $(S, \circ)$  with an identity element  $\epsilon$  meaning that for all values  $a \in S$  it holds that  $\epsilon \circ a = a = a \circ \epsilon$ . The identity element can be added to computation builder as the *Zero* member. This operation is used when a computation uses conditional without *else* branch. Thus we get:

$$m \{ \mathbf{if\ false\ then} \ cexpr_1 \} \equiv m \{ cexpr_2 \} \equiv m \{ cexpr_2 \ \mathbf{if\ false\ then} \ cexpr_1 \}$$

Although these are simple laws, they can be used to reason about list comprehensions. The associativity means that we can move a *cexpr*sub-expression of computation expression (that uses *yield!* repeatedly) into a separate computation. To use the identity law, consider a recursive function that generates numbers up to 100:

```
let rec range n = seq { yield n
                       if n < 100 then yield! range (n + 1) }
```

The law guarantees that for  $n = 100$ , the body is equivalent to  $m \{ \mathbf{yield} \ 100 \}$ . This is an expected property of the *if* construct – the law guarantees that it holds even for *if* that is reinterpreted by some (monoidal) computation expression.

### 4.2 Monad and additive monad laws

Monad laws are well-understood and the corresponding equivalent computation expressions do not significantly differ from the laws about Haskell’s *do* notation:

$$m \{ \mathbf{let!} \ y = m \{ \mathbf{return} \ x \} \ \mathbf{in} \ cexpr \} \equiv m \{ \mathbf{let} \ y = x \ \mathbf{in} \ cexpr \} \\ m \{ \mathbf{let!} \ x = c \ \mathbf{in} \ \mathbf{return} \ x \} \equiv m \{ \mathbf{return!} \ c \}$$

$$\begin{aligned} & \text{m } \{ \text{let! } x = \text{m } \{ \text{let! } y = c \text{ in } cexpr_1 \} \text{ in } cexpr_2 \} \equiv \\ & \equiv \text{m } \{ \text{let! } y = c \text{ in let! } x = \text{m } \{ cexpr_1 \} \text{ in } cexpr_2 \} \end{aligned}$$

**Resolving ambiguity.** When discussing the translation rules §3.3, we noted that the rules are ambiguous when both *Map* and *Bind* operations are present. The following can be translated both monadically and applicatively:

$$\text{m } \{ \text{let! } x = c \text{ in return } expr \}$$

The two translations are shown below. Assuming that our computation is a monad, this is a well-known definition of *Map* in terms of *Bind* and *Return*:

$$\text{m.Map}(x, \text{fun } x \rightarrow expr) \equiv \text{m.Bind}(x, \text{fun } x \rightarrow \text{m.Return}(expr))$$

More generally, if a computation builder defines both *Map* and *Bind* (even if they are not based on a monad), we require this equation to guarantee that the two possible translations produce equivalent computations.

**Additive monads.** Additive monads are computations that combine monad with the monoidal structure. As shown earlier §2.2, these can be embedded using *let!/return* or using *for/yield*. The choice can be made based on the laws that hold.

The laws required for additive monads is not fully resolved [8]. A frequently advocated law is *left distributivity* – binding on the result of a monoidal operation is equivalent to binding on two computations and then combining the results:

$$\text{m.For}(\text{m.Combine}(a, b), f) \equiv \text{m.Combine}(\text{m.For}(a, f), \text{m.For}(b, f))$$

We intentionally use the *For* operation (corresponding to the *for* keyword), because this leads to the following intuitive syntactic equality:

$$\text{m } \{ \text{for } x \text{ in m } \{ cexpr_1; cexpr_2 \} \text{ do } cexpr \} \equiv \text{m } \{ \text{for } x \text{ in m } \{ cexpr_1 \} \text{ do } cexpr \text{ for } x \text{ in m } \{ cexpr_2 \} \text{ do } cexpr \}$$

If we read the code as an imperative looping construct (without the computational reinterpretation), then this is, indeed, a valid law about *for* loops.

Another law that is sometimes required about additive monads is *left catch*. It states that combining a computation that immediately returns a value with any other computation results in a computation that just returns the value:

$$\text{m.Combine}(\text{m.Return}(v), a) \equiv \text{m.Return}(v)$$

This time, we intentionally used the *Return* member instead of *Yield*, because the law corresponds to the following syntactic equivalence:

$$\text{m } \{ \text{return } v; cexpr \} \equiv \text{m } \{ \text{return } v \}$$

The fact that *left catch* corresponds to an intuitive syntactic equality about *let!/return* while *left distributivity* corresponds to an intuitive syntactic equality about *for/yield* determines the appropriate syntax. The former can be used for list comprehensions (and other collections), while the latter is suitable e.g. for the option monad or the software transactional memory monad [6].

### 4.3 Monad transformers

There are multiple ways of composing or layering monads [11,13,3]. Monad transformers are perhaps the most widely known technique. A monad transformer is a type constructor  $Tm$  together with a *Lift* operation. For some monad  $M$  the operation has a type  $M\alpha \rightarrow TM\alpha$  and it turns a computation in the underlying monad into a computation in the composed monad.

The result of monad transformer is also a monad. This means that we can use the usual syntactic sugar for monads, such as the *do* notation in Haskell. However, a more specific notation can use the additional *Lift* operation.

We looked at computation expression for a composed monad when discussing asynchronous sequences §2.3. An asynchronous sequence `AsyncSeq  $\alpha$`  is a computation obtained by applying the list monad transformer [7] to the monad `Async  $\alpha$` . Asynchronous sequences are *additive monads* satisfying the left distributivity law, so we choose the *for*/*yield* syntax for working with the composed computation. We also provided an additional *Bind* operation to support awaiting a single asynchronous workflow using *let!* This operation is defined in terms of *Lift* of the monad transformer and *For* (monadic bind) of the composed computation:

$$\text{asyncSeq.Bind}(a, f) = \text{asyncSeq.For}(\text{asyncSeq.Lift}(a), f)$$

There are two laws that hold about monad transformers. To simplify the presentation, we use asynchronous workflows and sequences rather than showing the generalised version. The first law states that composing *Return* of asynchronous workflows with *Lift* should be equivalent to the *Yield* of asynchronous sequences. The other states that *Lift* distributes over monadic bind.

Our syntax always combines *Lift* with *For*, so the following syntactic equivalences also require right identity for monads and function extensionality:

$$\begin{aligned} \text{asyncSeq } \{ \text{let! } x = \text{async } \{ \text{return } v \} \text{ in return } x \} &\equiv \text{asyncSeq } \{ \text{return } v \} \\ \text{asyncSeq } \{ \text{let! } x = \text{async } \{ \text{let! } y = c \text{ in } cexpr_1 \} \text{ in } cexpr_2 \} &\equiv \\ &\equiv \text{asyncSeq } \{ \text{let! } y = c \text{ in let! } x = \text{async } \{ cexpr_1 \} \text{ in } cexpr_2 \} \end{aligned}$$

The first equation returns  $v$  without any asynchronous waiting in both cases (although, in presence of side-effects, this is made more complicated by cancellation). The second equation is more subtle. The left-hand side awaits a single asynchronous workflow that first awaits  $c$  and then does more work. The right-hand side awaits  $c$  lifted to an asynchronous sequence and then awaits the rest.

### 4.4 Applicative computations

The last type of computations that we discussed §2.4 is *applicative functor*. We use the less common definition called *Monoidal* [14]. It consists of *Map* and *Merge*, together with a unit computation. We use the unit to define *Zero* – it will be used in computations that contain some unit-returning expression, such as `()`.

The identity law guarantees that merging with a unit and then projecting the non-unit value produces an equivalent computation:

$$f \{ \mathbf{let!} \ x = f \{ () \} \ \mathbf{and} \ y = c \ \mathbf{in} \ \mathbf{return} \ y \} \} \equiv c \equiv f \{ \mathbf{let!} \ x = c \ \mathbf{and} \ y = f \{ () \} \ \mathbf{in} \ \mathbf{return} \ x \} \}$$

The naturality law specifies that *Merge* distributes over *Map*, which translates to the following code (assuming  $x_1$  not free in  $expr_2$  and  $x_2$  not free in  $expr_1$ ):

$$\begin{aligned} & f \{ \mathbf{let!} \ y_1 = f \{ \mathbf{let!} \ x_1 = c_1 \ \mathbf{in} \ \mathbf{return} \ expr_1 \} \ \mathbf{and} \ y_2 = f \{ \mathbf{let!} \ x_2 = c_2 \ \mathbf{in} \ \mathbf{return} \ expr_2 \} \ \mathbf{in} \ expr \} \equiv \\ & \equiv f \{ \mathbf{let!} \ x_1 = c_1 \ \mathbf{and} \ x_2 = c_2 \ \mathbf{in} \ \mathbf{let} \ y_1, y_2 = expr_1, expr_2 \ \mathbf{in} \ expr \} \end{aligned}$$

As with the earlier syntactic rules, we can leave out the non-standard aspect of the computations, read them as ordinary functional code and get correct and expected laws. This means that the laws, again, guarantee that intuition about the syntax used by computation expressions will be correct.

Finally, the *Merge* operation is also required to be associative – this does not have any corresponding syntax, but it means that the user does not need to know implementation details of the compiler – it does not matter whether the parsing of *binds* in *let!...and...in* is left-associative or right-associative.

## 5 Related work

Haskell and its extensions support monad comprehensions [13] and “do” notation for monads, idiom brackets [14] for applicatives and arrows [16]. These are similar to computation expressions in that they are not tied to concrete computations. However, they differ syntactically – they add multiple new notations, while computation expressions add a uniform notation resembling standard language structures. Adding arrows to computation expressions is an open question.

Python and C# generators, LINQ [15] in C# and “for” comprehensions in Scala are just a few examples of syntax for concrete computations. Although they can all be used with other computations, this is not generally considered idiomatic use. Similarly to F#, the Scala *async* library [22] supports loops and exception handling. However, it is implemented through full macro system.

Aside from monads, other ways of encoding effectful computations include effect handlers [18] and continuations [3]. Providing syntactic support for these encodings may provide an interesting alternative to our encoding using computation builder. Interestingly, our *Run* operation seems related to *reset* of delimited continuations [19] and our *Delay* is similar to the *reify* operation of Filinsky [4].

## 6 Conclusions

In this paper, we related F# *computation expressions* to well-known abstract computation types. Computation expressions provide a unified way for writing a wide range of computations including monoids, monads, applicative formlets and monads composed using monad transformers.

Computation expressions follow a different approach than e.g. Haskell “do” notation. They integrate a wide range of abstractions and flexibly reuse existing syntax (including loops and exception handling). The library developer can choose the appropriate syntax and use laws of abstract computations to guarantee that the computation preserves intuition about the syntax.

Such reusable syntactic extensions are becoming increasingly important. We cannot keep adding new features to support comprehensions, asynchronicity, queries and more as the “syntactic budget” is rapidly running out.

## References

1. G. Bierman, C. Russo, G. Mainland, E. Meijer, and M. Torgersen. Pause ‘n’ play: formalizing asynchronous c#. ECOOP, 2012.
2. E. Cooper, S. Lindley, P. Wadler, and J. Yallop. The essence of form abstraction. APLAS, 2008.
3. A. Filinski. Representing layered monads. POPL, pages 175–188, 1999.
4. A. Filinski. Monads in action. POPL, pages 483–494, 2010.
5. G. Giorgidze, T. Grust, N. Schweinsberg, and J. Weijers. Bringing back monad comprehensions. Haskell Symposium, pages 13–22, 2011.
6. T. Harris, S. Marlow, S. Peyton-Jones, and M. Herlihy. Composable memory transactions. PPOPP, pages 48–60, 2005.
7. HaskellWiki. Listt done right. Available at [http://www.haskell.org/haskellwiki/ListT\\_done\\_right](http://www.haskell.org/haskellwiki/ListT_done_right), 2012.
8. HaskellWiki. Monadplus. Available at <http://www.haskell.org/haskellwiki/MonadPlus>, 2012.
9. G. Hutton and E. Meijer. Monadic parsing in haskell. *J. Funct. Program.*, 8(4):437–444, July 1998.
10. S. Krishnamurthi. Artifact evaluation for software conferences. Available at <http://cs.brown.edu/~sk/Memos/Conference-Artifact-Evaluation/>, 2012.
11. S. Liang, P. Hudak, and M. Jones. Monad transformers and modular interpreters. POPL, 1995.
12. S. Lindley, P. Wadler, and J. Yallop. Idioms are oblivious, arrows are meticulous, monads are promiscuous. *Electron. Notes Theor. Comput. Sci.*, 229(5), Mar. 2011.
13. C. Lüth and N. Ghani. In *Proceedings of the seventh ACM SIGPLAN international conference on Functional programming*, ICFP, pages 133–144, 2002.
14. C. McBride and R. Paterson. Applicative programming with effects. *J. Funct. Program.*, 18(1):1–13, Jan. 2008.
15. E. Meijer, B. Beckman, and G. Bierman. LINQ: reconciling object, relations and XML in the .NET framework. SIGMOD, pages 706–706, 2006.
16. R. Paterson. A new notation for arrows. ICFP, 2001.
17. T. Petricek. Programming with f# asynchronous sequences. Available at <http://tomasp.net/blog/async-sequences.aspx>, 2011.
18. G. Plotkin and M. Pretnar. Handlers of algebraic effects. ESOP, 2009.
19. T. Ropmf, I. Maier, and M. Odersky. Implementing first-class polymorphic delimited continuations by a type-directed selective cps-transform. ICFP, 2009.
20. D. Syme, T. Petricek, and D. Lomov. The f# asynchronous programming model. PADL, 2011.
21. The F# Software Foundation. F# language specification. 2013.
22. Typesafe Inc. An asynchronous programming facility for scala. Available at <http://github.com/scala/async>, 2013.
23. P. Wadler. Monads for functional programming. In *Advanced Funct. Prog.*, 1995.